Example 8.5 Heat Transfer in a Rectangle
Equation (8.1.5) is solved in Maple below:

```
> restart:with(inttrans):with(plots):
```

The governing equation and boundary conditions are entered and converted to the Laplace domain.

```
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t}\,u(x,t) = \frac{\partial^2}{\partial x^2}\,u(x,t) \tag{1}$$

```
> u(x,0):=1;
```

$$u(x,0) := 1 \tag{2}$$

```
> bc1:=u(x,t)=0;
```

$$bc1 := u(x,t) = 0 \tag{3}$$

```
> bc2:=u(x,t)=0;
```

$$bc2 := u(x,t) = 0 \tag{4}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\,U(x) - 1 = \frac{d^2}{dx^2}\,U(x) \tag{5}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(diff(laplace(u(x,t),t,s),x)=D(U)(0),laplace(u(x,t),t,
  s)=U(0),bc1);
```

$$bc1 := U(0) = 0 \tag{6}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(diff(laplace(u(x,t),t,s),x)=D(U)(1),laplace(u(x,t),t,
  s)=U(1),bc2);
```

$$bc2 := U(1) = 0 \tag{7}$$

The solution obtained in the Laplace domain is:

```
> U(x):=rhs(dsolve({eqs,bc1,bc2},U(x)));
```

$$U(x) := -\frac{e^{\sqrt{s}\,x}\left(-1+e^{-\sqrt{s}}\right)}{s\left(-e^{\sqrt{s}}+e^{-\sqrt{s}}\right)} + \frac{e^{-\sqrt{s}\,x}\left(-1+e^{\sqrt{s}}\right)}{s\left(-e^{\sqrt{s}}+e^{-\sqrt{s}}\right)} + \frac{1}{s} \tag{8}$$

Maple fails to invert the solution obtained:

```
> invlaplace(U(x),s,t);
```

$$-\frac{1}{2}\,invlaplace\left(\frac{e^{\sqrt{s}\,x}}{s\sinh\left(\sqrt{s}\right)}, s, t\right) - invlaplace\left(\frac{e^{\sqrt{s}\,x-\sqrt{s}}}{s\left(-e^{\sqrt{s}}+e^{-\sqrt{s}}\right)}, s, t\right) + \frac{1}{2}\,invlaplace\left(\frac{e^{-\sqrt{s}\,x}}{s\sinh\left(\sqrt{s}\right)}, s, t\right) + invlaplace\left(\frac{e^{-\sqrt{s}\,x+\sqrt{s}}}{s\left(-e^{\sqrt{s}}+e^{-\sqrt{s}}\right)}, s, t\right) + 1 \tag{9}$$

The first two terms of U(x) are expressed as an infinite series below:

```
> U1s:=-exp(s^(1/2)*x)/s/(exp(s^(1/2))+1);
```

$$U1s := -\frac{e^{\sqrt{s}\,x}}{s\left(e^{\sqrt{s}}+1\right)} \tag{10}$$

```
> U2s:=-exp(-s^(1/2)*x)*exp(s^(1/2))/s/(exp(s^(1/2))+1);
```

$$U2s := -\frac{e^{-\sqrt{s}\,x}\,e^{\sqrt{s}}}{s\left(e^{\sqrt{s}}+1\right)} \tag{11}$$

We want to write a series in terms of S=exp(-s^(1/2)) so that the series will converge:

```
> U1S:=series(subs(exp(s^(1/2))=1/S,U1s),S);
```

$$U1S := -\frac{e^{\sqrt{s}\,x}}{s}\,S + \frac{e^{\sqrt{s}\,x}}{s}\,S^2 - \frac{e^{\sqrt{s}\,x}}{s}\,S^3 + \frac{e^{\sqrt{s}\,x}}{s}\,S^4 - \frac{e^{\sqrt{s}\,x}}{s}\,S^5 + \mathrm{O}\left(S^6\right) \tag{12}$$

```
> U1S:=subs(S=exp(-s^(1/2)),U1S);
```

$$U1S := -\frac{e^{\sqrt{s}\,x}\,e^{-\sqrt{s}}}{s} + \frac{e^{\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^2}{s} - \frac{e^{\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^3}{s} + \frac{e^{\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^4}{s} - \frac{e^{\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^5}{s} + \mathrm{O}\left(\left(e^{-\sqrt{s}}\right)^6\right) \tag{13}$$

```
> simplify(U1S);
```

$$\frac{-e^{\sqrt{s}\,(x-1)} + e^{\sqrt{s}\,(x-2)} - e^{\sqrt{s}\,(x-3)} + e^{\sqrt{s}\,(x-4)} - e^{\sqrt{s}\,(x-5)} + \mathrm{O}\left(e^{-6\sqrt{s}}\right)s}{s} \tag{14}$$

Hence, U1S can be written as the infinite series:

```
> U1S:=Sum((-1)^n*exp(s^(1/2)*(x-n))/s,n=1..infinity);
```

$$U1S := \sum_{n=1}^{\infty} \frac{(-1)^n\,e^{\sqrt{s}\,(x-n)}}{s} \tag{15}$$

The general term in the above series is:

```
> u1s:=(-1)^n*exp(s^(1/2)*(x-n))/s;
```

$$u1s := \frac{(-1)^n\,e^{\sqrt{s}\,(x-n)}}{s} \tag{16}$$

The time domain solution for this expression is:

```
> u1t:=invlaplace(u1s,s,t);
```

$$u1t := (-1)^n\,invlaplace\left(\frac{e^{\sqrt{s}\,(x-n)}}{s}, s, t\right) \tag{17}$$

Hence, the inverse of U1S is the infinite series given by:

```
> U1t:=Sum(u1t,n=1..infinity);
```

$$U1t := \sum_{n=1}^{\infty} (-1)^n\,invlaplace\left(\frac{e^{\sqrt{s}\,(x-n)}}{s}, s, t\right) \tag{18}$$

Similarly, U2S is inverted below:

```
> U2S:=series(subs(exp(s^(1/2))=1/S,U2s),S);
```

$$U2S := -\frac{e^{-\sqrt{s}\,x}}{s} + \frac{e^{-\sqrt{s}\,x}}{s} S - \frac{e^{-\sqrt{s}\,x}}{s} S^2 + \frac{e^{-\sqrt{s}\,x}}{s} S^3 - \frac{e^{-\sqrt{s}\,x}}{s} S^4 + \frac{e^{-\sqrt{s}\,x}}{s} S^5 + O(S^6) \quad (19)$$

```
> U2S:=subs(S=exp(-s^(1/2)),U2S);
```

$$U2S := -\frac{e^{-\sqrt{s}\,x}}{s} + \frac{e^{-\sqrt{s}\,x}\,e^{-\sqrt{s}}}{s} - \frac{e^{-\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^2}{s} + \frac{e^{-\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^3}{s} - \frac{e^{-\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^4}{s} + \frac{e^{-\sqrt{s}\,x}\left(e^{-\sqrt{s}}\right)^5}{s} + O\left(\left(e^{-\sqrt{s}}\right)^6\right) \quad (20)$$

```
> simplify(U2S);
```

$$\frac{-e^{-\sqrt{s}\,x} + e^{-\sqrt{s}\,(x+1)} - e^{-\sqrt{s}\,(x+2)} + e^{-\sqrt{s}\,(x+3)} - e^{-\sqrt{s}\,(x+4)} + e^{-\sqrt{s}\,(x+5)} + O\left(e^{-6\sqrt{s}}\right) s}{s} \quad (21)$$

```
> U2S:=Sum((-1)^n*exp(-s^(1/2)*(x+n-1))/s,n=1..infinity);
```

$$U2S := \sum_{n=1}^{\infty} \frac{(-1)^n e^{-\sqrt{s}\,(x+n-1)}}{s} \quad (22)$$

```
> u2s:=(-1)^n*exp(-s^(1/2)*(x+n-1))/s;
```

$$u2s := \frac{(-1)^n e^{-\sqrt{s}\,(x+n-1)}}{s} \quad (23)$$

```
> u2t:=invlaplace(u2s,s,t);
```

$$u2t := (-1)^n \, invlaplace\left(\frac{e^{(-x-n+1)\sqrt{s}}}{s}, s, t\right) \quad (24)$$

```
> U2t:=Sum(u2t,n=1..infinity);
```

$$U2t := \sum_{n=1}^{\infty} (-1)^n \, invlaplace\left(\frac{e^{(-x-n+1)\sqrt{s}}}{s}, s, t\right) \quad (25)$$

The final solution for u in the time domain is:

```
> Ut:=U1t+U2t+1;
```

$$Ut := \sum_{n=1}^{\infty} (-1)^n \, invlaplace\left(\frac{e^{\sqrt{s}\,(x-n)}}{s}, s, t\right) + \sum_{n=1}^{\infty} (-1)^n \, invlaplace\left(\frac{e^{(-x-n+1)\sqrt{s}}}{s}, s, t\right) + 1 \quad (26)$$

For plotting purposes, infinity is replaced by N = 20:

```
> u:=subs(infinity=N,Ut);
```

$$u := \sum_{n=1}^{N} (-1)^n \, invlaplace\left(\frac{e^{\sqrt{s}\,(x-n)}}{s}, s, t\right) + \sum_{n=1}^{N} (-1)^n \, invlaplace\left(\frac{e^{(-x-n+1)\sqrt{s}}}{s}, s, t\right) + 1 \quad (27)$$

```
> u:=subs(N=20,u);
```

(28)

$$u := \sum_{n=1}^{20} (-1)^n \mathit{invlaplace}\left(\frac{e^{\sqrt{s}\,(x-n)}}{s}, s, t\right) + \sum_{n=1}^{20} (-1)^n \mathit{invlaplace}\left(\frac{e^{(-x-n+1)\sqrt{s}}}{s}, s, t\right) + 1 \qquad \textbf{(28)}$$

The following plots can be obtained:

```
> plot3d(u,x=0..1,t=1e-6..0.1,axes=boxed,title="Figure 8.9.",
  labels=[x,t,"u"],orientation=[60,60]);
```

Figure 8.9.




```
> plot([subs(t=1e-6,u),subs(t=1e-3,u),subs(t=0.01,u),subs(t=0.05,
  u)],x=0..1,axes=boxed,title="Figure 8.10.",thickness=5,labels=
  [x,"u"]);
```

Figure 8.10.



Note that for plotting purposes t = 0 is replaced by $t = 10^{-6}$ to avoid singularity at t = 0.

>